```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the third Weyl algebra $A = A_3(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d[1],x[1]], diff=[d[2],x[2]],
> diff=[d[3],x[3]], polynom=[x[1],x[2],x[3]]):
```

and the left A-module L finitely presented by the following matrix:

```
> P := evalm([[x[2]*d[1]/2,x[2]*d[2]+1,x[2]*d[3]+d[1]/2],
> [-x[2]*d[2]/2-3/2,0,d[2]/2],[-d[1]-x[2]*d[3]/2,-d[2],-d[3]/2]]);
```

$$P := \begin{bmatrix} \frac{1}{2}\,x_2\,d_1 & x_2\,d_2+1 & x_2\,d_3+\frac{1}{2}\,d_1 \\ -\frac{1}{2}\,x_2\,d_2-\frac{3}{2} & 0 & \frac{1}{2}\,d_2 \\ -d_1-\frac{1}{2}\,x_2\,d_3 & -d_2 & -\frac{1}{2}\,d_3 \end{bmatrix}$$

Let us compute its rank:

```
> OreRank(P,A);
```

$$1$$

We obtain $rank_A(\mathrm{L}) = 1$. Hence, L is either a free left A-module of rank 1 or L can be generated by two elements.

Let us compute Stafford's reduction of L:

```
> S := StaffordReduction(P,A,"reduce_generators"=true);
```

$$S := \left[\begin{bmatrix} -\frac{1}{2}\,d_1\,x_2\,d_2-\frac{3}{2}\,d_1-\frac{1}{2}\,d_3\,{x_2}^2\,d_2-\frac{3}{2}\,x_2\,d_3 & \frac{1}{2}\,x_2\,d_3\,d_2+\frac{1}{2}\,d_2\,d_1 \\ -\frac{1}{2}\,x_2\,d_2-\frac{3}{2} & \frac{1}{2}\,d_2 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}\right]$$

We obtain that the left A-homomorphism γ from L_2 to L induced by S[2], where L_2 is the left A-module finitely presented by S[1], is a left A-isomorphism. This result can be checked again:

```
> TestIso(S[1],P,S[2],A);
```

$$true$$

Let us try to reduce the number of relations in the presentation matrix S[1] of L_2:

```
> T := StaffordReduction(S[1],A,"checkrank"=false,"reduce_generators"=true);
```

$$T := [\begin{bmatrix} 0 \end{bmatrix}, \begin{bmatrix} x_2 & -1 \end{bmatrix}]$$

We get that L_2 is isomorphic to the left A-module finitely presented by T[1], and thus isomorphic to A. This last isomorphism can be checked again:

```
> TestIso(T[1],S[1],T[2],A);
```

$$true$$

Thus, L is a free left A-module of rank 1. If we define U = T[2] S[2], i.e.,

```
>  U := Mult(T[2],S[2],A);
```

$$U := \left[\begin{array}{ccc} x_2 & 0 & -1 \end{array}\right]$$

then the left A-homomorphism θ from A to L induced by U is a left A-isomorphism:

```
>  TestIso(T[1],P,U,A);
```

$$true$$

In particular, since 1 is a basis of A, then $\theta(1)$ is a basis of L.